TEST

CONTENTS

1 FAMILY PLANNING INDICATORS

In Chapter 4, we'll demonstrate how to calculate key family planning indicators appearing in the **PMA Longitudinal Brief** for each of the longitudinal samples currently available from IPUMS PMA. The brief for each sample is linked below.

- Burkina Faso
- DRC - Kinshasa
- DRC - Kongo Central
- Kenya
- Nigeria - Kano
- Nigeria - Lagos

Chapter 5 includes code you can use to reproduce the **alluvial plots** seen in these briefs.

Indicators calculated in this chapter cover topics like:

- pregnancy intentions and outcomes
- current use of long-acting, short-acting, and traditional contraceptives
- discontinuation of family planning
- intentions for future use of family planning
- unmet need for family planning
- partner's support for use of family planning

As we demonstrate how to calculate these indicators, we'll also provide code you can use to check for statistically significant differences between subgroups and plot the results as a bar chart showing a 95% confidence interval for each estimate.

1.1 CHAPTER SETUP

Chapter 4 features a **wide** longitudinal extract with all 6 of the available samples. Unlike Chapter 3, the data extract used in this chapter includes **only Female Respondents**.



Using the variable selection process described in Chapter 2, add the following variables to your Data Cart and click the View Cart button to begin checkout (preselected variables are added automatically).

- resultfq - Result of female questionnaire
- panelweight - Phase 2 female panel weight
- resident - Household residence / membership
- age - Age in female questionnaire
- pregnant - Pregnancy status
- birthevent - Number of birth events
- educattgen - Highest level of school attended (4 categories)
- marstat - Marital status
- geocd - Province, DRC
- geong - State, Nigeria
- cp - Contraceptive user
- fpcurreffmethrc - Most effective current FP method
- unmetyn - Total unmet need
- fppartsupport - Husband / partner would be supportive of FP use
- fpplanval - When will start using FP method in the future - value
- fpplanwhen - When will start using FP method in the future - unit
- country - PMA country (preselected)
- eaid - Enumeration area (preselected)

Before completing checkout, make sure that you've selected the Stata data format.



Download your data extract and load it into Stata:

```
use "pma_00121.dta", clear
```

1.2 SURVEY DESIGN

We've mentioned in Chapter 1 that PMA samples are only valid for the *de facto* population: these are women who slept in the household during the night before the interview for the Household Questionnaire in both phases. These women are coded either `11` or `22` in both `RESIDENT_1` and `RESIDENT_2`. We'll drop all other women from our extract:

```
keep if inlist(resident_1,11,22) & inlist(resident_2,11,22)
```

We also mentioned in Chapter 2 that women who completed the Phase 1 Female Questionnaire may have been **lost to follow-up** at Phase 2. As a reminder, we'll need to drop any cases where `RESULTFQ_2` is not coded `1` for "fully completed".

```
keep if resultfq_2 == 1
```

Additionally, a small number of women in each sample elected not to respond to key questions regarding current use of contraceptives, reported in the variable cp. These cases are coded `90` and above, as shown on the cp Codes tab. We'll exclude those cases, as well.

```
keep if cp_1 < 90 & cp_2 <90
```

Finally, recall that only the Burkina Faso and Kenya samples are **nationally representative**. Samples from DRC represent regions identified by geocd, while samples from Nigeria represent regions identified by geong. In order to distinguish each population of interest, we'll define a custom variable `pop` that shows each sample's country label concatenated with each of these regions where appropriate.

```
gen pop = .
replace pop = 1 if country == 1 // Burkina Faso
replace pop = 2 if country == 2 & geocd == 1 // Kinshasa
replace pop = 3 if country == 2 & geocd == 2 // Kongo Central
replace pop = 4 if country == 7 // Kenya
replace pop = 5 if country == 9 & geong == 4 // Kano
replace pop = 6 if country == 9 & geong == 2 // Lagos

label define pop ///
           1 "Burkina Faso" ///
           2 "DRC-Kinshasa" ///
           3 "DRC-Kongo Central" ///
           4 "Kenya" ///
           5 "Nigeria-Kano" ///
           6 "Nigeria-Lagos", replace

label values pop pop
```

The remaining sample size for each population of interest is simply a count of each level in `pop`.

```
table ( pop ) ( ) ( ), nototals missing
```

```
---------------------------------
                    |  Frequency
--------------------+------------
pop                 |
  Burkina Faso      |      5,207
  DRC-Kinshasa      |      1,967
  DRC-Kongo Central |      1,511
  Kenya             |      6,934
  Nigeria-Kano      |        998
  Nigeria-Lagos     |      1,088
---------------------------------
```

Our data extract includes samples from the DRC - Kinshasa and DRC - Kongo Central, so we'll use the procedure described in Chapter 1 to create a variable for sample strata called `strata_recode`. To review: `strata_recode` uses unique numeric codes from `strata_1`, except that it also includes unique identifiers for each sampled region in `geocd`.[1]

```
clonevar strata_recode = strata_1
replace  strata_recode = geocd if country == 2
label copy STRATA_1 strata_recode, replace
label define strata_recode 1 "Kinshasa, DRC" 2 "Kongo Central, DRC", modify
label values strata_recode strata_recode
```

We'll now use the `svyset` command to incorporate survey design information into each of the population estimates calculated throughout this chapter. In addition to `strata_recode`, this includes survey weights provided by panelweight and sample clusters identified by eaid_1.[2]

```
svyset eaid_1, strata(strata_recode) weight(panelweight)
```

[1] Because women are considered "lost to follow-up" if they moved outside the study area, `strata_1` and `strata_2` are identical for all panel members: you can use either one to identify sample clusters.

[2] Because women are considered "lost to follow-up" if they moved outside the study area, `eaid_1` and `eaid_2` are identical for all panel members: you can use either one to identify sample clusters.

1.3 POPULATION INFERENCE

The `svyset` command passes the information in panelweight, eaid_1, and pop to other commands that use the `svy:` prefix, like `svy: tab` or `svy: proportion` or `svy: mean`. We'll also demonstrate how to use this information in formal significance tests within each sample via `svy: tab`.

Let's begin with a simple example. The variable cp indicates whether a woman was currently using any family planning method. The variables `cp_1` and `cp_2` in our **wide** extract represent responses collected at Phase 1 and Phase 2, respectively. With help from `svyset`, we'll obtain a population-level estimate of the proportion of women who were using a method at Phase 2, given their status at Phase 1.

Below, we use the `over(pop)` option to divide the data extract into individual samples defined by `pop`. We then use `svy: proportion` to generate population estimates obtained from the combined data extract that are identical to those you would obtain if you downloaded one extract for each sample and analyzed them separately.

You may change the confidence interval to, for example, 99% by setting the option `level(99)` in `svy: proportion`.

```
// Phase 2 status among women not using contraceptives in Phase 1
svy: proportion cp_2 if cp_1 == 0 , over(pop)
```

```
Survey: Proportion estimation

Number of strata =   6                       Number of obs   =     10,573
Number of PSUs   = 664                       Population size = 11,093.988
                                             Design df       =        658

--------------------------------------------------------------------------
                        |             Linearized            Logit
                        | Proportion   std. err.     [95% conf. interval]
------------------------+-------------------------------------------------
              cp_2@pop  |
      no Burkina Faso   |   .7898736   .0132519      .7626719    .8147149
      no DRC-Kinshasa   |   .7392914    .025396      .6864473    .7860059
 no DRC-Kongo Central   |   .7361748   .0243823      .6856104     .781202
             no Kenya   |   .6966579   .0109966      .6746419    .7178072
      no Nigeria-Kano   |   .9456453   .0146418      .9086278     .968191
     no Nigeria-Lagos   |   .7570627   .0205668      .7144437    .7951447
     yes Burkina Faso   |   .2101264   .0132519      .1852851    .2373281
     yes DRC-Kinshasa   |   .2607086    .025396      .2139941    .3135527
yes DRC-Kongo Central   |   .2638252   .0243823       .218798    .3143896
            yes Kenya   |   .3033421   .0109966      .2821928    .3253581
     yes Nigeria-Kano   |   .0543547   .0146418       .031809    .0913722
    yes Nigeria-Lagos   |   .2429373   .0205668      .2048553    .2855563
--------------------------------------------------------------------------
```

```
// Phase 2 status among women using contraceptives in Phase 1
svy: proportion cp_2 if cp_1 == 1 , over(pop)
```

```
Survey: Proportion estimation

Number of strata =   6                       Number of obs   =      7,132
Number of PSUs   = 654                       Population size = 6,597.2713
                                             Design df       =        648

--------------------------------------------------------------------------
                        |             Linearized            Logit
                        | Proportion   std. err.     [95% conf. interval]
------------------------+-------------------------------------------------
              cp_2@pop  |
      no Burkina Faso   |   .3473058   .0217125      .3059791     .391069
      no DRC-Kinshasa   |   .2747554   .0188858      .2392609    .3133467
 no DRC-Kongo Central   |   .2696023   .0342418      .2078232    .3418219
             no Kenya   |   .1996438   .0086007      .1832896    .2170694
      no Nigeria-Kano   |   .4399902   .0665671      .3161123     .571823
     no Nigeria-Lagos   |   .2397128   .0233586      .1968771    .2885204
     yes Burkina Faso   |   .6526942   .0217125       .608931    .6940209
     yes DRC-Kinshasa   |   .7252446   .0188858      .6866533    .7607391
yes DRC-Kongo Central   |   .7303977   .0342418      .6581781    .7921768
            yes Kenya   |   .8003562   .0086007      .7829306    .8167104
     yes Nigeria-Kano   |   .5600098   .0665671       .428177    .6838877
    yes Nigeria-Lagos   |   .7602872   .0233586      .7114796    .8031229
--------------------------------------------------------------------------
```

The population estimate for each row appears in the column `Proportion`. Looking at row 1 in the first table, we would estimate that 79% of women aged 15-49 in Burkina Faso used *no method* both at Phase 1 and again at Phase 2. The columns below `[95% conf. interval]` list the limits of a two-sided 95% Logit confidence interval: 76.3% and 81.5%.

Note that each population appears twice in each table: in the top half of the table we see the estimated proportion who *were not* using contraception in Phase 2, and in the bottom half we see the complementary proportions – those who *were* using contraception in Phase 2. Each proportion in the top half corresponds to a proportion in the bottom half, and each pair sums to 1.0 (or 100%). For example, the 79% of women in Burkina Faso shown in row 1 of the first table sums with 21% in row 7 to equal 100%.

Because these tables include complementary proportions, it is not necessary to plot both the top and bottom half of each table. Instead, we'll use negative space to help the reader visualize complementary proportions in a **grouped bar chart**.

1.3.1 Data Visualization

We'll use simple **grouped bar charts** to show population estimates for each proportion calculated throughout the remainder of this chapter. We'll also include **error bars** representing a 95% confidence interval for each proportion.

For example, let's consider how to visualize the two tables produced by `svy: proportion` in the previous section. It is possible to make a barchart in Stata starting with what we might call *raw data* in memory using the `graph bar` command. It may be combined with `aweight` and `by` and `over` options to include a weighted sub-graph for each country and each level of Phase 1 contraception use. But it will not show confidence intervals, so in this chapter we take a different approach and use the more versatile `graph twoway` family of commands to build up the figure that we want. Specifically, we use `twoway bar` `twoway rcap` to plot bars and confidence intervals, respectively.

But for `graph twoway` we cannot plot directly from the raw data. We need to construct a new dataset with one row per bar in the bar chart that stores the survey estimated proportions and confidence interval limits or bounds. There are two ways to construct that dataset: a) using stored output from commands like those shown in the previous section,[3] or b) by writing a short program to construct the dataset row-by-row.

Throughout this chapter, we use the second approach as shown below. Here we construct the plotting dataset, one country at a time, one dataset row at a time, and focusing only on the proportion of women who were using contraception in Phase 2. We accomplish this with a program that uses nested for-loops and the `svy: proportion` command to calculate the estimate and confidence interval for each bar in our chart and, while those coordinates are in memory, uses the `post` command to write them to a new dataset.

A full tutorial on the family of post commands is beyond the scope of this chapter – they are quite useful and worth taking some time to understand – the Stata documentation for them is here.

First, we prepare to capture the summary dataset as `postout`.

```
capture postclose toplot
tempfile postout
postfile toplot cp_1 cp_2 pop estimate lcb ucb using `postout', replace
```

[3] In addition to display in the Stata log window, the table of results obtained above is stored by Stata for optional downstream use. The .do-file that accompanies this book includes code to access the results and wrangle them into a dataset to use for plotting. (The results are stored as a matrix that may be brought into memory, but needs to be transposed and needs to have several ID variables re-instated before it is ready for plotting, so there are several data management steps involved.)

Then, we loop through each value of `cp_1` (`i`) for each `pop` (`j`). Within each loop, we extract the estimates from column 2 of `r(table)` because we are summarizing the proportion who were using contraception in Phase 2, which means we want to know the proportion of "yes" responses (`1`) in `cp_2`.

```
forvalues i = 0/1 {
    forvalues j = 1/6 {
        svy, subpop(if cp_1 == `i' & pop == `j') : proportion cp_2
        post toplot (`i') (1) (`j') ///
                    (`=100*r(table)[1,2]') /// // the estimate
                    (`=100*r(table)[5,2]') /// // the LCB
                    (`=100*r(table)[6,2]')     // the UCB
    }
}
capture postclose toplot
use `postout', clear

label define yesno 0 "No" 1 "Yes", replace
label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values cp_1 yesno
label values cp_2 yesno
label values pop pop
label variable cp_1 "Contraceptive user (Phase 1)"
label variable cp_2 "Contraceptive user (Phase 2)"
```

Next we use Stata's `twoway` command to make a grouped bar chart, with mostly default aesthetic options.

```
* Basic graph
twoway (bar estimate cp_1, horizontal  ///
        ylabel(0(1)1, valuelabel ) ///
        xlabel(0(20)100)) ///
        (rcap lcb ucb cp_1 , horizontal ) ///
        ,  by(pop, legend(off) ) ///
           xtitle(Contraceptive user (Phase 2) (%))
```



And finally, we incorporate some additional syntax to bring more aspects of the aesthetics under our control:

```
* Additional aesthetic options
twoway (bar estimate cp_1 if cp_1 == 1 & cp_2 == 1, ///
          color(blue*.5) horizontal ylabel(0(1)1,valuelabel angle(0) nogrid) ///
          xlabel(0(20)100)) ///
       (bar estimate cp_1 if cp_1 == 0 & cp_2 == 1, ///
          color(orange*.5) horizontal) ///
       (rcap lcb ucb cp_1 if cp_2 == 1, horizontal lcolor(black)) ///
       ,  by(pop, graphregion(color(white)) legend(off) note("")) ///
          subtitle(,lcolor(white) fcolor(white)) ///
          xtitle(Contraceptive user (Phase 2) (%)) ///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          name(nicer1, replace)
```



1.3.2 Significance Test

Comparing confidence intervals in the bar chart shown above gives us an informal, *conservative* way to test for a significant difference between outcomes for each `pop`: because none of the confidence intervals overlap within any given population, we can be at least 95% confident that the proportion of Phase 2 contraceptive users is not identical for Phase 1 users compared with Phase 1 non-users.

Formal testing may also reveal significant differences between pairs of outcomes where these intervals overlap slightly. Our approach is well suited for data visualization, but it should not replace formal testing. Fortunately, you can adapt our code to replace (or complement) the output from `svy: proportion`.

For example, we'll now use a Rao-Scott chi-square test for significant differences between the estimated population proportions for Burkina Faso and the proportions we would *expect* to observe if Phase 2 outcomes were statistically independent from Phase 1 conditions.[4]

```
svy, subpop(if pop == 1): tab cp_1 cp_2 , row ci nomarginals pearson null
```

```
Number of strata =   1                          Number of obs   =      5,207
Number of PSUs   = 167                          Population size = 5,215.6413
                                                Subpop. no. obs =      5,207
                                                Subpop. size    = 5,215.6413
                                                Design df       =        166


----------------------------------------
Contracep |
tive user | Contraceptive user (Phase 2)
(Phase 1) |            no            yes
----------+-----------------------------
       no |         .7899          .2101
          | [.7625,.8148]  [.1852,.2375]
          |
      yes |         .3473          .6527
          | [.3058,.3913]  [.6087,.6942]
----------------------------------------
Key: Row proportion
     [95% confidence interval for row proportion]

  Pearson:
    Uncorrected   chi2(1)         =  934.8348
    D-B (null)    F(1, 166)       =  305.9783     P = 0.0000
    Design-based  F(1, 166)       =  472.5643     P = 0.0000

Note: 5 strata omitted because they contain no subpopulation members.
```

[4] See Stata help for `svy: tab` and the references there for more information.

The p-value for the Rao-Scott test is in the row of output labeled `D-B (null)` (where the null hypothesis is that the proportions are equal to the product of the marginal probabilities). We see a p-value of `0.0000`, so we reject the null hypothesis that the proportion of Phase 2 contraceptive users might be the same regardless of contraceptive use at Phase 1.

You can perform a separate test for each `pop` by constructing a for-loop like the one shown below. This will generate one table for each population in `pop` (results omitted for space constraints).

```
forvalues i = 1/6 {
    di "Study Population: `: label pop `i''"
    svy, subpop(if pop == `i'): tab cp_1 cp_2 , row ci nomarginals pearson null
}
```

Suppose instead that we wanted to know whether a significant difference exists between Phase 1 users in Burkina Faso compared with those from DRC-Kongo. It's hard to tell from our bar chart whether the confidence intervals overlap in this case, but we can easily adapt our Rao-Scott test to check for a difference between populations.

```
svy: tab cp_2 pop if inlist(pop,1,3) & cp_1 == 1, null pearson col
```

```
Number of strata =   2                        Number of obs   =      2,412
Number of PSUs   = 223                        Population size = 2,054.7818
                                              Design df       =        221


----------------------------------------
Contracep |
tive user |             pop
(Phase 2) |  Burkina  DRC-Kong     Total
----------+-----------------------------
       no |    .3473     .2696     .3264
      yes |    .6527     .7304     .6736
          |
    Total |        1         1         1
----------------------------------------
Key: Column proportion

  Pearson:
    Uncorrected   chi2(1)         =   13.0465
    D-B (null)    F(1, 221)       =    3.3293     P = 0.0694
    Design-based  F(1, 221)       =    3.4134     P = 0.0660
```

This time, the value in `D-B (null)` is `0.0694`, so we fail to reject the null hypothesis. The likelihood that a difference exists between Phase 1 users in Burkina Faso and DRC-Kongo is less than 95%.

1.4 CONTRACEPTIVE USE OR NON-USE

Let's continue our examination of cp. In the PMA reports for each sample linked above, you'll notice that women who were pregnant at either phase are distinguished from women who reported use or non-use in `CP_1` or `CP_2`. We'll identify these women in the variable pregnant, and then we'll create a combined indicator called `fpstatus`.

- `fpstatus` - Pregnant, using contraception, or using no contraception

We'll create `fpstatus_1` to incorporate pregnancy information into the contraceptive use status of women at Phase 1, and `fpstatus_2` at Phase 2.

```
gen fpstatus_1 = 1 if pregnant_1 == 1
replace fpstatus_1 = 3 if pregnant_1 != 1 & cp_1 == 1
replace fpstatus_1 = 2 if pregnant_1 != 1 & cp_1 == 0

gen fpstatus_2 = 1 if pregnant_2 == 1
replace fpstatus_2 = 3 if pregnant_2 != 1 & cp_2 == 1
replace fpstatus_2 = 2 if pregnant_2 != 1 & cp_2 == 0

label define status 1 "Pregnant" 3 "Using FP" 2 "Not Using FP"
label values fpstatus_1 status
label values fpstatus_2 status
```

Examining the first dozen rows of the dataset below, we see that `fpstatus` is "Pregnant" for pregnant women, and indicates contraceptive use otherwise.

```
list pregnant_1 cp_1 fpstatus_1 pregnant_2 cp_2 fpstatus_2 in 1/12, noobs sep(12)
```

```
+-------------------------------------------------------------------+
| pregna~1   cp_1     fpstatus_1   pregna~2   cp_2     fpstatus_2 |
|-------------------------------------------------------------------|
|       no     no   Not Using FP         no    yes       Using FP |
|      yes     no       Pregnant         no    yes       Using FP |
|       no     no   Not Using FP         no     no   Not Using FP |
|       no    yes       Using FP         no     no   Not Using FP |
|       no     no   Not Using FP         no     no   Not Using FP |
|       no     no   Not Using FP         no     no   Not Using FP |
|       no     no   Not Using FP         no     no   Not Using FP |
|       no     no   Not Using FP         no     no   Not Using FP |
|       no    yes       Using FP         no    yes       Using FP |
|       no     no   Not Using FP         no    yes       Using FP |
|       no    yes       Using FP         no    yes       Using FP |
|       no    yes       Using FP         no     no   Not Using FP |
+-------------------------------------------------------------------+
```

We'll now use these new variables to address questions like:

- Are women who were pregnant at Phase 1 more likely to use or not use family planning at Phase 2?
- Are women who were using (or not using) contraception at Phase 1 likely to maintain the same status at Phase 2?

We can modify the earlier approach to building a dataset for making a grouped bar chart. First, we prepare a new dataset `postout`.

```
capture postclose toplot
tempfile postout
postfile toplot fpstatus_1 fpstatus_2 pop estimate lcb ucb using `postout', replace

forvalues i = 1/3 {
    forvalues k = 1/3 {
        forvalues j = 1/6 {
            capture drop y
            gen y = fpstatus_2 == `k'
            svy, subpop(if fpstatus_1 == `i' & pop == `j'): proportion y
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define status 1 "Pregnant" 3 "Using FP" 2 "Not Using FP"
label values fpstatus_1 status
label values fpstatus_2 status
label define pop 1 "Burkina Faso" 2 "DRC–Kinshasa" 3 "DRC–Kongo Central" ///
  4 "Kenya" 5 "Nigeria–Kano" 6 "Nigeria–Lagos", replace
label values pop pop

label variable fpstatus_1 "Family Planning Status (Phase 1)"
label variable fpstatus_2 "Family Planning Status (Phase 2)"
```

And next, we build a bar chart with customized aesthetic options.

```
label define status2 1 "Pregnant in Phase 1" 3 "Using FP in Phase 1" 2 "Not Using FP in
Phase 1", replace
label values fpstatus_1 status2
twoway (bar estimate fpstatus_2 if fpstatus_2 == fpstatus_1, ///
        color(blue*.5) horizontal ///
        ylabel(1(1)3,valuelabel angle(0) nogrid) ///
        xlabel(0(20)100)) ///
        (bar estimate fpstatus_2 if fpstatus_2 != fpstatus_1, ///
        color(orange*.5) horizontal) ///
        (rcap lcb ucb fpstatus_2 , horizontal lcolor(black)) ///
        ,  by(pop fpstatus_1, graphregion(color(white)) ///
            note("") col(3)  ) ///
           subtitle(,lcolor(white) fcolor(white)) ///
           xtitle(Family Planning Status (Phase 2)) ///
           xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
           legend(order(1 "No change" 2 "Status changed") size(vsmall) ///
                  region(lcolor(white)) symxsize(small) symysize(small)) ///
           ytitle("")
```



Estimates within each subplot sum to 100%.

To reiterate: comparing the error bars within each of these 18 panels gives us a informal, but conservative test for significant difference. We'll say that a significant difference occurs where two pairs of error bars **do not overlap** (but additional testing may be necessary to determine whether a significant difference occurs where error bars overlap only slightly). A few observations:

- For women who were pregnant at Phase 1, there is usually no apparent difference between using and not using family planning at Phase 2. Kenya and Nigeria - Kano are the exception: in Kenya, pregnant women at Phase 1 were appear more likely to be using FP at Phase 2, while the opposite is true in Kano.
- Overall, non-pregnant women at Phase 1 appeared more likely to maintain the same status (use or non-use) at Phase 2 than they were to switch or become pregnant.

1.5 CONTRACEPTIVE METHOD TYPE

PMA surveys also ask contraceptive users to indicate which method they are currently using at each phase of the study. If a woman reports using more than one method, fpcurreffmeth shows her most *effective* currently used method. These responses are combined with detailed information about use of the lactational amenorrhea method (LAM), emergency contraception, or injectable type in fpcurreffmethrc. PMA reports use fpcurreffmethrc to determine whether each woman's most effective current method is a short-acting, long-acting, or traditional method.

Long-acting methods include:

- IUDs
- implants
- male sterilization
- female sterilization

Short-acting methods include:

- injectables (intramuscular and subcutaneous)
- the pill
- emergency contraception
- male condoms
- female condoms
- LAM
- diaphragm
- foam/jelly
- standard days method

Traditional methods include:

- rhythm
- withdrawal
- other traditional

These methods are coded sequentially by group in fpcurreffmethrc. Women who are “NIU (not in universe)” were using no method.

```
table ( fpcurreffmethrc_1 ) ( ) ( ), nototals missing
```

```
-------------------------------------------------------------------
                                                  |  Frequency
--------------------------------------------------+----------------
most effective current fp method (numeric, recoded) |
  female sterilization                            |        198
  male sterilization                              |          1
  implants                                        |      2,248
  iud                                             |        226
  injectables (3 months)                          |      1,412
  injectables (sayana press)                      |        296
  pill                                            |        547
  emergency contraception                         |        243
  male condom                                     |        791
  female condom                                   |          1
  diaphragm                                       |          1
  foam                                            |          1
  standard days/cycle beads method                |         70
  lactational amenorrhea method (lam)             |         24
  rhythm                                          |        569
  withdrawal                                      |        351
  other traditional                               |        153
  no response or missing                          |          1
  niu (not in universe)                           |     10,572
-------------------------------------------------------------------
```

We’ll recode the Phase 1 and Phase 2 versions of fpcurreffmethrc into three groups representing short-acting, long-acting, and traditional methods.

```
* Generate new variables to recode the methods to 3 categories
label define fpmethod 4 "Long-acting" 3 "Short-acting" 2 "Traditional" 1 "None", replace

foreach v in fpcurreffmethrc_1 fpcurreffmethrc_2 {
    gen cat_`v' = 4 if `v' < 120
    replace cat_`v' = 3 if `v' >= 120 & `v' < 200
    replace cat_`v' = 2 if `v' >= 200 & `v' < 900
    replace cat_`v' = 1 if cat_`v' == .

    label values cat_`v' fpmethod
}
```

Next, we'll generate population estimates for our recoded variables.

```
capture postclose toplot
tempfile postout
postfile toplot methcat_1 methcat_2 pop estimate lcb ucb using `postout', replace

forvalues i = 1/4 {
    forvalues k = 1/4 {
        forvalues j = 1/6 {
            capture drop y
            gen y = cat_fpcurreffmethrc_2 == `k'
            svy, subpop(if cat_fpcurreffmethrc_1 == `i' & pop == `j'): proportion y
            svy: proportion y if cat_fpcurreffmethrc_1 == `i' & pop == `j'
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define fpmethod 4 "Long-acting" 3 "Short-acting" 2 "Traditional" 1 "None", replace
label values methcat_1 fpmethod
label values methcat_2 fpmethod

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop

label variable methcat_1 "Family Planning Method (Phase 1)"
label variable methcat_2 "Family Planning Method (Phase 2)"
```

And finally, we plot the results.

```
twoway (bar estimate methcat_2 if methcat_1 == methcat_2, ///
          color(blue*.5) horizontal ///
          ylabel(1(1)4,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
       (bar estimate methcat_2 if methcat_1 != methcat_2, ///
          color(orange*.5) horizontal) ///
       (rcap lcb ucb methcat_2 , horizontal lcolor(black)) ///
       ,  by(pop (methcat_1), graphregion(color(white)) ///
             note(Graphs by population and Phase 1 method, size(vsmall)) col(4)) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle(Family Planning Method (Phase 2), size(small)) ///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          legend(order(1 "No change" 2 "Method changed") size(vsmall) ///
                 region(lcolor(white)) symxsize(small) symysize(small)) ///
          ytitle("")
```



Estimates within each subplot sum to 100%.
Graphs by population and Phase 1 method.

What do we learn from this bar chart? Let's consider each column in turn:

- Users of "long-acting" methods at Phase 1 appear more likely to have used "long-acting" methods at Phase 2 than to have changed status (except perhaps in Kano, where the intervals for "long-acting" and "none" overlap at Phase 2).
- Users of "short-acting" methods at Phase 1 appeared generally likely to use them again at Phase 2, but some samples show that women are equally likely to be using "none" at Phase 2. A difference between these two outcomes is visually apparent only in Kinshasa, Kenya, and Lagos (where women were more likely to be using "short-acting" methods than "none").
- The status of Phase 1 "traditional" users is generally unclear at Phase 2. In Kinshasa, Kongo Central, and Lagos, these women seem most likely to remain "traditional" users at Phase 2. Elsewhere, there are no clear trends.
- Users of "none" at Phase 1 were clearly most likely to remain as such at Phase 2.

1.6 CONTRACEPTIVE DYNAMICS BY SUBGROUP

We can also use fpcurreffmethrc to see whether women switched methods, stopped using any method, started using any method, or made no changes. Let's summarize this information as `chg_fpcurr`:

- `chg_fpcurr` - Change in contraceptive use between Phase 1 and Phase 2

```
gen chg_fpcurr = .
replace chg_fpcurr = 1 if fpcurreffmethrc_1 < 900 & ///
  fpcurreffmethrc_2 < 900 & ///
  fpcurreffmethrc_1 != fpcurreffmethrc_2
replace chg_fpcurr = 2 if fpcurreffmethrc_1 < 900 & ///
  fpcurreffmethrc_2 < 900 & ///
  fpcurreffmethrc_1 == fpcurreffmethrc_2
replace chg_fpcurr = 3 if fpcurreffmethrc_1 > 900 & fpcurreffmethrc_2 > 900
replace chg_fpcurr = 4 if fpcurreffmethrc_1 > 900 & fpcurreffmethrc_2 < 900
replace chg_fpcurr = 5 if fpcurreffmethrc_1 < 900 & fpcurreffmethrc_2 > 900

label define chg_fpcurr 1 "Changed methods" 2 "Continued method" ///
  3 "Continued non-use" 4 "Started using" 5 "Stopped using", replace
label values chg_fpcurr chg_fpcurr
label var chg_fpcurr = "Phase 1 to 2 Family Planning Change Status"
```

PMA reports disaggregate the outcomes captured in `chg_fpcurr` by age, marital status, education level, and parity (number of live childbirths).

1.6.1 Age

We'll use PMA's categorization of `age_2` to examine differences between women in three categories in `cat_age_2`.

```
gen cat_age_2 = .
replace cat_age_2 = 1 if age_2 < 20
replace cat_age_2 = 2 if age_2 >= 20 & age_2 < 25
replace cat_age_2 = 3 if age_2 >= 25
label define cat_age_2 1 "15-19" 2 "20-24" 3 "25-49", replace
label values cat_age_2 cat_age_2
label var cat_age_2 "Age category at Phase 2"

capture postclose toplot
tempfile postout
postfile toplot cat_age_2 chg_fpcurr pop estimate lcb ucb using `postout', replace

forvalues i = 1/3 {
    forvalues k = 1/5 {
        forvalues j = 1/6 {
            capture drop y
            gen y = chg_fpcurr == `k'
            svy, subpop(if cat_age_2 == `i' & pop == `j'): proportion y
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define cat_age_2 1 "15-19" 2 "20-24" 3 "25-49", replace
label define chg_fpcurr 1 "Changed methods" 2 "Continued method" ///
  3 "Continued non-use" 4 "Started using" 5 "Stopped using", replace

label values cat_age_2 cat_age_2
label values chg_fpcurr chg_fpcurr

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop

label var chg_fpcurr = "Phase 1 to 2 Family Planning Change Status"
```

Plotting `cat_age_2` on the y-axis allows us to compare confidence intervals across age groups. For example, notice that women aged 15-19 in every population seem more likely to continue non-use than women who are aged 20-24 or 25-49 (column 3).

```
twoway (bar estimate cat_age_2 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(1(1)3,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
       (rcap lcb ucb cat_age_2 , horizontal lcolor(black)) ///
       ,  by(pop chg_fpcurr, graphregion(color(white)) ///
             note("") col(5) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Changes in Family Planning Status, Phase 1 to Phase 2", size(small))
///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("")
```



Changes in Family Planning Status, Phase 1 to Phase 2

Estimates across each row of the figure sum to 100%.

1.6.2 Education level

The variable educattgen standardizes educational categories across countries.[5] To match PMA reports, we'll recode educattgen into just three groups in `cat_educattgen_2`.

```
gen cat_educattgen_2 = .
replace cat_educattgen_2 = 1 if educattgen_2 < 3
replace cat_educattgen_2 = 2 if educattgen_2 == 3
replace cat_educattgen_2 = 3 if educattgen_2 == 4
label define cat_educattgen_2 1 "None/Primary" 2 "Secondary" 3 "Tertiary", replace
label values cat_educattgen_2 cat_educattgen_2
label var cat_educattgen_2 "Education Category at Phase 2"

capture postclose toplot
tempfile postout
postfile toplot cat_educattgen_2 chg_fpcurr pop estimate lcb ucb using `postout', replace

forvalues i = 1/3 {
    forvalues k = 1/5 {
        forvalues j = 1/6 {
            capture drop y
            gen y = chg_fpcurr == `k'
            svy, subpop(if cat_educattgen_2 == `i' & pop == `j'): proportion y
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define cat_educattgen_2 1 "None/Primary" 2 "Secondary" 3 "Tertiary", replace
label define chg_fpcurr 1 "Changed methods" 2 "Continued method" ///
  3 "Continued non-use" 4 "Started using" 5 "Stopped using", replace

label values cat_educattgen_2 cat_educattgen_2
label values chg_fpcurr chg_fpcurr

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop

label var chg_fpcurr = "Phase 1 to 2 Family Planning Change Status"
```

[5] See educatt for country-specific codes.

As with age, we'll plot `cat_educattgen_2` on the y-axis. There aren't many clear takeaways here: confidence intervals overlap in each column for almost every education level, so visual inspection reveals no clear significant differences.

```
twoway (bar estimate cat_educattgen_2 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(1(1)3,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
       (rcap lcb ucb cat_educattgen_2 , horizontal lcolor(black)) ///
       ,  by(pop chg_fpcurr, graphregion(color(white)) ///
             note("") col(5) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Changes in Family Planning Status, Phase 1 to Phase 2", size(small))
///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("")
```

Changes in Family Planning Status, Phase 1 to Phase 2

Estimates across each row of the figure sum to 100%.

In Nigeria-Lagos, the confidence intervals for Secondary and Tertiary overlap, but not by much, so it may be interesting to examine the outcome of a formal Rao-Scott chi-square test.

```
capture drop y
gen y = chg_fpcurr == 3
svy, subpop(if inlist(cat_educattgen_2,2,3) & pop == 6): ///
     tab cat_educattgen_2 y if cat_educattgen_2 > 1 , row pearson null ci
```

```
(running tabulate on estimation sample)

Number of strata =  1                           Number of obs   =        970
Number of PSUs   = 51                           Population size = 957.828852
                                                Subpop. no. obs =        970
                                                Subpop. size    = 957.828852
                                                Design df       =         50

-------------------------------------------------------
Education |
Category  |
at Phase  |                      y
2         |             0              1          Total
----------+--------------------------------------------
 Secondar |         .5088          .4912              1
          | [.4601,.5573]  [.4427,.5399]
          |
 Tertiary |         .5892          .4108              1
          | [.5331,.6431]  [.3569,.4669]
          |
    Total |         .5434          .4566              1
          | [.5058,.5805]  [.4195,.4942]
-------------------------------------------------------
Key: Row proportion
     [95% confidence interval for row proportion]

  Pearson:
    Uncorrected   chi2(1)         =    6.1953
    D-B (null)    F(1, 50)        =    5.3040     P = 0.0255
    Design-based  F(1, 50)        =    5.3259     P = 0.0252

Note: 5 strata omitted because they contain no subpopulation members.
```

This is an example where even though the confidence intervals overlap, the chi-square test rejects the null hypothesis that the proportion of respondents who continued non-use is equal in the secondary versus tertiary category in Lagos, with a p-value of 0.0255.

1.6.3 Marital status

The variable marstat indicates each woman's marital / partnership status. PMA considers women "in union" to be those who are currently married (code 21) or currently living with their partner (code 22). Otherwise, women who were never married, divorced / separated, or widowed are considered "not in union". We'll assign these values to `cat_marstat_2`.

```
gen cat_marstat_2 = .
replace cat_marstat_2 = 1 if marstat_2 == 21 | marstat_2 == 22
replace cat_marstat_2 = 2 if cat_marstat_2 != 1
label define cat_marstat_2 1 "In union" 2 "Not in union", replace
label values cat_marstat_2 cat_marstat_2
label variable cat_marstat_2 "Marital status at Phase 2"

capture postclose toplot
tempfile postout
postfile toplot cat_marstat_2 chg_fpcurr pop estimate lcb ucb using `postout', replace

forvalues i = 1/2 {
    forvalues k = 1/5 {
        forvalues j = 1/6 {
            capture drop y
            gen y = chg_fpcurr == `k'
            svy, subpop(if cat_marstat_2 == `i' & pop == `j'): proportion y
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define cat_marstat_2 1 "In union" 2 "Not in union", replace
label define chg_fpcurr 1 "Changed methods" 2 "Continued method" ///
  3 "Continued non-use" 4 "Started using" 5 "Stopped using", replace

label values cat_marstat_2 cat_marstat_2
label values chg_fpcurr chg_fpcurr

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop

label var chg_fpcurr "Phase 1 to 2 Family Planning Change Status"
```

Here, we see that women who were *not* in a union at Phase 2 were significantly more likely to continue non-use of contraception compared to married / partnered women in each population. On the other hand, women who *were* in a union mainly appeared more likely to continue using the same method, or perhaps to change methods (most clearly in Kenya).

```
twoway (bar estimate cat_marstat_2 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(1(1)2,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
        (rcap lcb ucb cat_marstat_2 , horizontal lcolor(black)) ///
        ,  by(pop chg_fpcurr, graphregion(color(white)) ///
             note("Estimates across each row of the figure sum to 100%.", ///
                    size(vsmall)) col(5) legend(off)) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Changes in Family Planning Status, Phase 1 to Phase 2", size(small))
///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("")
```



1.6.4 Parity

Parity refers to the number of times a women has given live birth (excluding stillbirths). This information is recorded in the IPUMS variable birthevent, in which the values `0` and `99` (not in universe) can both be interpreted as "none". We'll create `cat_birthevent_2` for Phase 2 parity in four groups.

```
gen cat_birthevent_2 = .
replace cat_birthevent_2 = 1 if inlist(birthevent_2,0,99)
replace cat_birthevent_2 = 2 if inlist(birthevent_2,1,2)
replace cat_birthevent_2 = 3 if inlist(birthevent_2,3,4)
replace cat_birthevent_2 = 4 if birthevent_2 >= 5 & birthevent_2 < 90
label define cat_birthevent_2 1 "None" 2 "One-two" 3 "Three-four" 4 "Five +", replace
label values cat_birthevent_2 cat_birthevent_2
label var cat_birthevent_2 "Parity (number of live births) at Phase 2"

capture postclose toplot
tempfile postout
postfile toplot cat_birthevent_2 chg_fpcurr pop estimate lcb ucb using `postout', replace

forvalues i = 1/4 {
    forvalues k = 1/5 {
        forvalues j = 1/6 {
            capture drop y
            gen y = chg_fpcurr == `k'
            svy, subpop(if cat_birthevent_2 == `i' & pop == `j'): proportion y
            post toplot (`i') (`k') (`j') ///
                        (`=100*r(table)[1,2]') /// // the estimate
                        (`=100*r(table)[5,2]') /// // the LCB
                        (`=100*r(table)[6,2]')     // the UCB
        }
    }
}
capture postclose toplot
use `postout', clear

label define cat_birthevent_2 1 "None" 2 "One-two" 3 "Three-four" 4 "Five +", replace
label define chg_fpcurr 1 "Changed methods" 2 "Continued method" ///
  3 "Continued non-use" 4 "Started using" 5 "Stopped using", replace

label values cat_birthevent_2 cat_birthevent_2
label values chg_fpcurr chg_fpcurr

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop

label var chg_fpcurr "Phase 1 to 2 Family Planning Change Status"
```

There are few clear patterns related to parity, except that women who have never given birth are also more likely to continue non-use of contraception between phases.

```
twoway (bar estimate cat_birthevent_2 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(1(1)4,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
        (rcap lcb ucb cat_birthevent_2 , horizontal lcolor(black)) ///
        ,  by(pop chg_fpcurr, graphregion(color(white)) ///
             note("Estimates across each row of the figure sum to 100%.", ///
                    size(vsmall)) col(5) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Changes in Family Planning Status, Phase 1 to Phase 2", size(small))
///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("")
```



1.7 OUTCOMES FOR PHASE 1 NON-USERS

The final page in each PMA report covers family planning dynamics related to unmet need, partner support, and plans for future use of family planning methods. In each case, we'll be focusing on women who were *not* using any method at Phase 1. We'll show how each of these dynamics impacts the likelihood that Phase 1 non-users would have adopted any family planning method at Phase 2.

1.7.1 Unmet need

PMA defines unmet need for family planning according to each woman's fertility preferences, current use of family planning methods, and risk factors for pregnancy. Women may have "unmet need" for birth spacing (e.g. pregnant women whose pregnancy was mistimed) or for limiting births (e.g. pregnant women whose pregnancy was unwanted), while women are considered "not at risk" if they are not sexually active or cannot become pregnant.

The variable unmetneed provides detailed information on types of need for each woman, and on related variables that were used to calculate unmet need. The binary variable unmetyn recodes unmetneed as either "Unmet need", or "No unmet need".

```
keep if cp_1 == 0

capture postclose toplot
tempfile postout
postfile toplot unmetyn_1 pop estimate lcb ucb using `postout', replace

forvalues i = 0/1 {
    forvalues j = 1/6 {
        capture drop y
        gen y = cp_2 == 1
        svy, subpop(if unmetyn_1 == `i' & pop == `j'): proportion y
        post toplot (`i') (`j') ///
                    (`=100*r(table)[1,2]') /// // the estimate
                    (`=100*r(table)[5,2]') /// // the LCB
                    (`=100*r(table)[6,2]')     // the UCB
    }
}
capture postclose toplot
use `postout', clear

label define unmetyn_1 0 "No" 1 "Yes", replace
label values unmetyn_1 unmetyn_1

label define pop 1 "Burkina Faso" 2 "DRC-Kinshasa" 3 "DRC-Kongo Central" ///
  4 "Kenya" 5 "Nigeria-Kano" 6 "Nigeria-Lagos", replace
label values pop pop
```

Overall, our bar chart suggests that non-users with unmet need for family planning at Phase 1 were more likely to adopt a method at Phase 2 compared to non-users who had none (e.g. women who were not sexually active, could not become pregnant, etc.). However, formal testing is needed to determine whether these trends were statistically significant in Burkina Faso and Nigeria - Kano.

```
twoway (bar estimate unmetyn_1 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(0(1)1,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
       (rcap lcb ucb unmetyn_1 , horizontal lcolor(black)) ///
       ,  by(pop , graphregion(color(white)) ///
             note("Among women who were not using family planning in Phase 1.", ///
                  size(vsmall)) col(1) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Had adopted a method of family planning in Phase 2", size(small)) ///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("Had unmet need in Phase 1") ///
          ysize(10) xsize(8)
```



Here, we conduct a Rao-Scott chi-square test for Burkina Faso:

```
* Women who didn't use any method at Phase 1
keep if cp_1 == 0

* Outcome is women using contraception in Phase 2
gen y = cp_2 == 1
label variable y "Using contraception in Phase 2"
label variable unmetyn_1 "Unmet Need in Phase 1"

* Test for difference in Burkina Faso
svy, subpop(if pop == 1 ): tab unmetyn_1 y, row pearson null ci
```

```
(running tabulate on estimation sample)

Number of strata =   1                          Number of obs   =      3,410
Number of PSUs   = 167                          Population size = 3,714.9751
                                                Subpop. no. obs =      3,410
                                                Subpop. size    = 3,714.9751
                                                Design df       =        166

------------------------------------------------------------
Unmet     |
Need in   |       Using contraception in Phase 2
Phase 1   |             0              1          Total
----------+-------------------------------------------------
 no unmet |         .8056          .1944              1
          | [.7743,.8334]  [.1666,.2257]
          |
 unmet ne |          .752           .248              1
          | [.7125,.7878]  [.2122,.2875]
          |
    Total |         .7899          .2101              1
          | [.7625,.8148]  [.1852,.2375]
------------------------------------------------------------
Key: Row proportion
     [95% confidence interval for row proportion]

  Pearson:
    Uncorrected   chi2(1)         =   12.2006
    D-B (null)    F(1, 166)       =    7.0687     P = 0.0086
    Design-based  F(1, 166)       =    7.4473     P = 0.0070

Note: 5 strata omitted because they contain no subpopulation members.
```

And here, we conduct a Rao-Scott chi-square test for Nigeria-Kano:

```
* Test for difference in Nigeria-Kano
svy, subpop(if pop == 5 ): tab unmetyn_1 y, row pearson null ci
```

```
(running tabulate on estimation sample)

Number of strata =  1                                  Number of obs   =     881
Number of PSUs   = 25                                  Population size = 901.912
                                                       Subpop. no. obs =     881
                                                       Subpop. size    = 901.912
                                                       Design df       =      24

------------------------------------------------------------
Unmet     |
Need in   |       Using contraception in Phase 2
Phase 1   |             0              1          Total
----------+-------------------------------------------------
 no unmet |         .9613          .0387              1
          | [.9294,.9791]  [.0209,.0706]
          |
 unmet ne |          .894           .106              1
          | [.7569,.9581]  [.0419,.2431]
          |
    Total |         .9456          .0544              1
          | [.9062,.9691]  [.0309,.0938]
------------------------------------------------------------
Key: Row proportion
     [95% confidence interval for row proportion]

  Pearson:
    Uncorrected   chi2(1)         =   13.8741
    D-B (null)    F(1, 24)        =    2.8789     P = 0.1027
    Design-based  F(1, 24)        =    5.8252     P = 0.0238

Note: 5 strata omitted because they contain no subpopulation members.
```

The output indicates that the difference **is** statistically significant in Burkina Faso (p=0.0086) and **is not** significant in Kano (p=0.1027).

1.7.2 Partner support

Women who were not using family planning and not pregnant at Phase 1 were asked whether they thought their husband / partner would be supportive of use of family planning in the future. These results are recorded in fppartsupport. In addition to women who were already using a method at Phase 1, we'll also exclude non-partnered women here, as they are "NIU (not in universe)".

```
keep if cp_1 == 0 & inlist(fppartsupport_1,0,1,97)
```

We'll recode the value 97 ("Do not know") in `fppartsupport_1` to the unused value of 2 because the value will also serve as the y-coordinate when we make the figure, and values of 0, 1, and 2 will be spaced in a more eye-pleasing manner than 0, 1, and 97.

```
replace fppartsupport_1 = 2 if fppartsupport_1 == 97

capture postclose toplot
tempfile postout
postfile toplot fppartsupport_1 pop estimate lcb ucb using `postout', replace

forvalues i = 0/2 {
    forvalues j = 1/6 {
        capture drop y
        gen y = cp_2 == 1
        svy, subpop(if fppartsupport_1 == `i' & pop == `j'): proportion y
        post toplot (`i') (`j') ///
                    (`=100*r(table)[1,2]') /// // the estimate
                    (`=100*r(table)[5,2]') /// // the LCB
                    (`=100*r(table)[6,2]')     // the UCB
    }
}
capture postclose toplot
use `postout', clear

label define fppartsupport_1 0 "No" 1 "Yes" 2 `"Do not know"', replace
label values fppartsupport_1 fppartsupport_1

label define pop 1 "Burkina Faso" 2 "DRC–Kinshasa" 3 "DRC–Kongo Central" ///
  4 "Kenya" 5 "Nigeria–Kano" 6 "Nigeria–Lagos", replace
label values pop pop
```

In our bar chart, Phase 2 adoption outcomes for women who answered “Do not know” are not visually distinct from those who answered “Yes” or “No”. Formal testing is needed to determine whether any significant differences exist.

Setting aside women who answered “Do not know”, women with Phase 1 partner support in DRC - Kongo Central and Kenya (“Yes”) were more likely to adopt a method than those without (“No”). Outcomes for women in other populations are not visibly different based on partner support, one way or the other (again, formal testing may prove otherwise).

```
twoway (bar estimate fppartsupport_1 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(0(1)2,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
        (rcap lcb ucb fppartsupport_1 , horizontal lcolor(black)) ///
        ,  by(pop , graphregion(color(white)) ///
             note("Among women who were not using family planning in Phase 1.", ///
                  size(vsmall)) col(1) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Had adopted a method of family planning in Phase 2", size(small)) ///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("Partner supportive of family planning in Phase 1") ///
          ysize(10) xsize(8)
```



1.7.3 Intentions

Lastly, we'll demonstrate the impact of women's plans for future family planning use at Phase 1. The variable fpusplan indicates whether women had plans for future use *at any point* in the future, but here we'll consider whether women had plans to adopt a method *within the next year* to correspond with the timing of Phase 2 surveys.

There are two variables that describe the approximate time when women said they would adopt a family planning method (if at all). fpplanval contains a raw number that should be matched with a *unit* of time (months, years) or a categorical response ("soon / now", "after the birth of this child") in fpplanwhen. We'll create `fpplanyr_1` to indicate whether each woman planned to use family planning within a year's time at Phase 1.

```
keep if cp_1 == 0

gen fpplanyr_1 =  (fpplanval_1  <= 12 & fpplanwhen_1 == 1) | ///
                  (fpplanval_1 == 1   & fpplanwhen_1 == 2) | ///
                  inlist(fpplanwhen_1,3,4)
label define fpplanyr_1 0 "No" 1 "Yes", replace
label values fpplanyr_1 fpplanyr_1
label var fpplanyr_1 "Plan to start using family planning within 1 year at Phase 1"

capture postclose toplot
tempfile postout
postfile toplot fpplanyr_1 pop estimate lcb ucb using `postout', replace

forvalues i = 0/1 {
    forvalues j = 1/6 {
        capture drop y
        gen y = cp_2 == 1
        svy, subpop(if fpplanyr_1 == `i' & pop == `j'): proportion y
        post toplot (`i') (`j') ///
                    (`=100*r(table)[1,2]') /// // the estimate
                    (`=100*r(table)[5,2]') /// // the LCB
                    (`=100*r(table)[6,2]')     // the UCB
    }
}
capture postclose toplot
use `postout', clear

label define fpplanyr_1 0 "No" 1 "Yes", replace
label values fpplanyr_1 fpplanyr_1

label define pop 1 "Burkina Faso" 2 "DRC–Kinshasa" 3 "DRC–Kongo Central" ///
  4 "Kenya" 5 "Nigeria–Kano" 6 "Nigeria–Lagos", replace
label values pop pop
```

In every population, Phase 1 non-users who planned to adopt a method by Phase 2 were significantly more likely to do so. However, only in Kenya do we see a significant *majority* of Phase 1 non-users with plans to adopt a method actually doing so, where the 95% confidence interval for "Yes" responses includes **only** proportions greater than the 50% threshold. In fact, women who adopted a method at Phase 2 represent a significant *minority* of Phase 1 non-users who planned to do so in Burkina Faso, DRC - Kongo Central, and Nigeria - Kano; in those populations, the entire 95% confidence interval falls below the 50% mark.

```
twoway (bar estimate fpplanyr_1 , ///
          color(blue*.5) horizontal barwidth(0.9) ///
          ylabel(0(1)1,valuelabel angle(0) nogrid labsize(small)) ///
          xlabel(0(20)100)) ///
        (rcap lcb ucb fpplanyr_1 , horizontal lcolor(black)) ///
        ,  by(pop , graphregion(color(white)) ///
             note("Among women who were not using family planning in Phase 1.", ///
                  size(vsmall)) col(1) legend(off) ) ///
          subtitle(,size(small) lcolor(white) fcolor(white)) ///
          xtitle("Had adopted a method of family planning in Phase 2", size(small)) ///
          xline(20 40 60 80 100, lcolor(gs15) lwidth(vthin)) ///
          ytitle("Respondent plans to adopt a FP method within a year at Phase 1") ///
          ysize(10) xsize(8)
```